


Financial Modelling in Excel (Intermediate)

Online Course Notes

Module 5 of 6

“Building a Business Case”



plumsolutions
Specialists in Financial Modelling

Contents of this Module:

MODULE FIVE: Building a Business Case	4
Example Case Study Model.....	4
Inputs for Example Case Study Model.....	4
Video 5.1: Business Case - Calculating Staff Costs.....	6
Video 5.2: Business Case - Adding Compounding Inflation	8
Adding Exponents to the Inflation Formula	8
Analysing Staff Costs.....	8
Error Checks in our Business Case	9
Video 5.3: Business Case - Forecasting Customer Numbers.....	10
Creating the Pool of Potential Customers.....	10
FORECASTing Number of Customers.....	10
Index Customers to Population Increase	10
Narrowing Down the Pool.....	11
Expected Customer Acquisition.....	11
Video 5.4: Business Case - Forecasting Customer Revenue.....	13
Churn (Customer Defection)	13
Premium Calculation.....	13
Video 5.5: Business Case - Modelling Market Penetration & Expenses	15
Modelling Market Penetration.....	15
Calculating Fixed & Variable Expenses	15
Video 5.6: Business Case - Project Evaluation (NPV & IRR)	17
Using a Goal Seek in our Business Case	17

Video 5.7: Business Case - Project Evaluation (Payback Period).....	19
Simple Payback Calculation.....	19
More Complex Payback Calculation	20

Important Notice: *These course materials are protected by copyright. They are provided to students of Plum Solutions Online Training courses in soft PDF format and so may be printed for study purposes by the licensed student, but may not be copied, redistributed, posted to a website or reproduced in any way.*

MODULE FIVE: Building a Business Case

Module 5 consists of 7 videos:

1. Business Case - Calculating Staff Costs
2. Business Case - Adding Compounding Inflation
3. Business Case - Forecasting Customer Numbers
4. Business Case - Forecasting Customer Revenue
5. Business Case - Modelling Market Penetration
6. Business Case - Project Evaluation (NPV & IRR)
7. Business Case - Project Evaluation (Payback Period)

Log into your online course and download the Excel templates on each video page. Follow along with the video demonstration, and save your model. Note that the Excel models in Module Two are not required uploads for this course, however, you are welcome to submit them if you'd like us to check that you're on the right track.

Example Case Study Model

You are required to build a business case for a new product. The exercise is as close to a real example as possible, and is very similar to the kinds of problems that you would face in the real world as a financial modeller.

The problem: Your company, "HoneyCorp", is considering starting a new pet insurance product and marketing it to existing customers, but you are dubious about whether it will make money or not. Build a business case to give the NPV, IRR and payback period, and calculate market penetration.

You are provided with the following pieces of information:

- Expected premium income
- Staffing costs
- Overheads project costs
- Potential customer pool
- Expected customer take-up rates

Inputs for Example Case Study Model

Based on the information given below, build a business case in Excel which will give the five year NPV and IRR and the payback period, and market penetration.

1. The planned annual premium is \$400 per pet. These premiums will increase by 3% inflation each year.
2. According to your data analysts, there are currently 2.2 million existing or past HoneyCorp customers in Australia.
3. It is expected that HoneyCorp will retain market share, and therefore we can assume that customer numbers will grow in line with the general population which, according to the Australian Bureau of Statistics, is expected to grow by around 1.5%. Grow customer numbers in line with population growth.
4. You do some research which tells you that 54% of Australian households own pets.
5. After prolonged discussions with the marketing team, it has been agreed that the expected take-up rate of existing customers who are pet owners will be 2.5% per annum.
6. You are not told anything about churn (lost customers) so make an educated assumption on churn.
7. Finance tells you that the profit margin on the premiums are around 10%. The rest is retained by the underwriter.
8. This margin does not include overheads (e.g. call centre staff, admin etc.) which are around **\$4.50** per customer per annum. It is assumed that this does not increase due to improved efficiency.
9. You are told that the media / marketing budget will be **\$500k** in Year 1, halving until Year 4 and then remaining constant. One-off product development costs will be **\$500k**.
10. HR have emailed you a list of charge out rates below for the staff you will need for the project.
11. The project manager has told you an approximate start and finish date for each resource, listing on the "Staff" page.
12. Using the Australian population projections below, calculate the projected market penetration and HoneyCorp customer base penetration of this product.
13. Calculate the 5 Year NPV, IRR and payback period for this project.
14. Hurdle rate or cost of capital for NPV is **12%**.
15. Create three scenarios; base case, worst case and best case. The base case data is as above, and for the worst case, take-up rates will drop to 2% and churn will increase by a few percentage points. Under the best case scenario, take-up rates will increase to 3% and churn will decrease by a few percentage points.

Video 5.1: Business Case - Calculating Staff Costs

For a project like this, staff costs are going to be the largest cost in your project, and the area of the model we will probably spend most time on.

1. Open the file “HC Business Case” as directed
2. Copy the staff data from the “Problem” page to the relevant section at the bottom of the “Assumptions” page as shown in the video
3. Make sure that your assumption have been formatted in **blue font** or using the “input” style on the Home tab under Styles. As we have discussed in previous modules, this is a kind of industry standard in financial modelling; a “signal” to other modellers that this is an input variable which can be changed.
4. Make sure you **document** your assumptions and sources. Where did you get the data from? What’s included? Do the costs include training, oncosts, recruitment etc? (yes they do!) List out what is included underneath the table.
5. The table doesn’t depict what time frame is covered. Is it a month? That’s a lot of money! It’s actually an **annual** rate, so we need to state that at the top.

See the section on “LOOKUP Fundamentals” in Module 2, as well as page 127 for more information on how to use the VLOOKUP function.

See the section on “Mixed References and Named Ranges” in Module 2, as well as page 123 of “Using Excel for Business Analysis” for how to create a Named Range.

6. Highlight this table on the “Assumptions” page and create a named range called “chargeout”.
7. Go to the “Staff” page of the model and in column D, match the staffing costs on the “Source Data” page of each employee using a **VLOOKUP** function. Make sure you ask it to return an exact match.
8. Your formula should be **=VLOOKUP(C7,chargeout,2,0)**. Drag the fill handle to copy the formula down.

Now we need to calculate how much we will spend in each year for each staff member. We’ll build the formula step by step. The best approach is to start with a simple formula, check that it is correct before moving onto the next step and making it more complex.

Remember, one of the principles of financial modelling is in a block of data to have the same formula throughout the entire block of data or table. Therefore, we’ll start in cell G7, and copy it across and down.

1. On the “Staff” sheet in cell G7, insert the formula **=IF(G6>=E7,D7,0)**. Add in your mixed referencing. Test it and make sure it is working correctly.
2. Now use an IF and AND nested function to calculate the annual staff costs of this project in

See the section on “Logical Nested Functions” in Module 2 of “Introduction to Financial Modelling” as well as page 112 of “Using Excel for Business Analysis” for more information on how to use Logical functions such as IF and AND nested together.

each year from column G. Insert the correct absolute referencing. In cell G7, your formula should look like this:

=IF(AND(G\$6>=\$E7,G\$6<=\$F7),\$D7,0) Copy this down and across the block of data again.

3. Check it, test it, and make sure it's right. Use your F2 shortcut key to "spot check" and make sure it's picking up the correct cells.
4. Then check your totals against those in the video before moving on

Video 5.2: Business Case - Adding Compounding Inflation

Now, we need to add inflation to the staff costs. One of the tricky things about calculating these salaries is that increases are based on the current amount, not their salary when they first started. We cannot simply link to the prior year, as our start and finish dates would no longer work. We need to leave the formula as it is, but add the compounding inflation at the end in stages, checking as we go.

Adding Exponents to the Inflation Formula

1. Create a named range called “inflation” on the “Assumptions” page, and enter 3%
2. Go back to the “Staff” page. We need to add inflation to this, so add $*(1+\text{inflation})$ to the end of the formula in **G7**. Your formula in cell **G7** should be:
=IF(AND(G\$6>=\$E7,G\$6<=\$F7),\$D7,0)*(1+inflation)
3. Inflation is compounding, which makes this a little trickier so an easy way to calculate for future years is to use the ^ function (“to the power of”). A common way of doing this would be to use the difference between the years as the escalation factor.
4. The formula in cell **G7** should be
=IF(AND(G\$6>=\$E7,G\$6<=\$F7),\$D7,0)*(1+inflation)^(G\$6-\$G\$6). Take special care with your dollar signs in the last part of the formula.
5. Make sure you have used correct absolute referencing, and then copy the formula all the way across and down the entire table, making sure that we have exactly the same formula throughout the whole table (so following financial modelling best practice).
6. Then check your totals against those in the video before moving on

Analysing Staff Costs

1. In our business case model, we’d like to know now how many people we are engaging for each position, so in cells E31:E41 create a formula which will count the number of times each position appears above. Insert a COUNTIF function (again using the wizard) in cell E31 next to the first name. Specify the range as column C in the table above, and the criteria as the name to the left of the cell. **=COUNTIF(C7:C26,C31)** This will count the number of people who hold that position.
2. Now, use absolute referencing to fix the rows and columns so that you can copy it down.
=COUNTIF(\$C\$7:\$C\$26,C31)
3. Check that the total number of staff in the table equal the total number of staff above. (i.e. 20) Now, in the area G31:P41 we need to summarise the amount spent on each position for each year.
4. Insert a SUMIF function in column G next to the first name. Specify the range as column C, the criteria as the name in column C, and the sum range as column G above.
=SUMIF(C7:C26,C31,G7:G26) This will sum the total cost for each position

See the section on “Aggregation Functions” in Module 2 as well as page 130 of Using Excel for Business Analysis for more detailed instructions on how to create aggregation functions such as SUMIF and COUNTIF.

5. Now, use absolute referencing to fix the rows and columns so that you can copy it down and across. Fix the references such that the sum range will copy across **=SUMIF(\$C\$7:\$C\$26,\$C31,G\$7:G\$26)**. Take care with your dollar signs!
6. Check your total by summing the data table at the top, and comparing it to your summary table at the bottom.

Error Checks in our Business Case

In our model, we know that both rows 27 and 42 should remain equal. We need to make sure they do even if someone else makes adjustments to the model.

1. The simplest kind of error check is to simply take one cell away from the other. Below the table, in cell **G43**, enter the formula **=G42-G27**.
2. Format it using the comma style and change the font colour to red
3. Alternatively, if you'd like to see the word "error", then use the formula **=IF(ABS(G42-G27)>0.1,"error",0)** instead.
4. Copy the formula across the entire row
5. Now apply conditional formatting to our error check such that if any cell returns an error you really can't miss it. Make the entire cell red with white font if the cell contains the word "error".

See the section on "Building Error Checks" in Module 3, as well as page 87 of Using Excel for Business Analysis for more detailed instructions on incorporating error checks in financial models.

See the section on "Conditional Formatting" in Module 3, as well as page 193 of Using Excel for Business Analysis for more detailed instructions on how to use conditional formatting in a financial model.

6. Now that we have finished with the staffing calculations, we need to pull this through to the summary. Go back to your "P&L" Summary Sheet, and link it through. Your formula in cell **B15** on the "P&L" sheet should be either **=Staff!G27** or **=Staff!G42**. (both rows contain the staff costs)

Video 5.3: Business Case - Forecasting Customer Numbers

The marketing team has told us that they expect take-up rates of 2.5% of the existing customer base. In reality, we would probably make rate decrease over time due to market saturation; however, for simplicity we have adopted a flat take-up rate. Firstly, we need to calculate the total pool of potential customers which are the HoneyCorp customers who are pet owners.

Creating the Pool of Potential Customers

1. Create a named range called "HC_Cust" on the "Assumptions" page and insert the number of customers. Link this to the HoneyCorp Customers section on the "Revenue" page.
2. We need to index the growth of HoneyCorp customers to the growth of the general population. On the "Assumptions" sheet in row 24, paste the general population forecast from the "Problem" sheet.
3. Document the source.
4. Oops, we seem to be missing a couple of years' data. We'll need to forecast.

See the section on "Assumptions Documentation Methods" in Module 5, as well as page 59 of Using Excel for Business Analysis for more detail about Assumptions Documentation.

FORECASTing Number of Customers

Note that you could simply highlight the historical data and copy it across, but it would be better practice to insert a formula instead so that other users can see where the numbers came from. If you simply copy it across, the numbers will be hard-coded, and users will assume they are an assumption unless documented otherwise.

1. Use an =FORECAST formula to forecast the last two years
2. On the "Assumptions" page, in cell J24, your formula should be: =FORECAST(J23,\$B\$24:\$I\$24,\$B\$23:\$I\$23)
3. Copy it across

See the section on "Using FORECAST / TREND functions" in Module 2, as well as page 161 of Using Excel for Business Analysis for more detailed instructions on using the FORECAST or TREND function.

Index Customers to Population Increase

1. Now, once we have all the population numbers, calculate the annual population increase using the formula =(C24-B24)/B24 and copy it across.
2. We are going to use these percentage increases to increase the projected number of customers. A key **assumption** of course, is that market share remains that same, and that if the total population increases, our customers will increase. Document this on your "Assumptions" page before we forget!
3. Now link this cell to the "Revenue" page and index the number of HoneyCorp customers to the growth in general population. Your formula should look like this =B3+(B3*Assumptions!C25). Copy it across.

Narrowing Down the Pool

Although all these people are potential customers, we know that the only ones who might take up our product are pet owners. Therefore we narrow down the pool to pet-owning HoneyCorp customers.

We do some research (see next page) which tells us that 54% of Australians own pets. Our customers are not any different to the general population, so it is reasonable to assume that the same pet-owning assumptions will apply.

1. Create a named range called “pet_ownshp” on the “Assumptions” page and enter 54%.
2. Document the source, and that we are assuming that our customers are represented by the total population.
3. Calculate pet owning HoneyCorp customers on the “Revenue” page in cell **B4** using the formula **=B3*pet_ownshp**
4. Copy it across

Expected Customer Acquisition

Based on market research, and take-up rates of similar products, we have been told to assume that 2.5% of the pet-owning HoneyCorp customers will take up this product. Note that this is 2.5% of the potential customers each and every year. Because the customers pay their premiums on an annual basis, we'll need to calculate the cumulative number of customers in order to work out the revenue.

1. Create a named range called “takeup” on the “Assumptions” page and enter 2.5%.
2. Calculate expected customer acquisition on the “Revenue” page in cell **B5** using the formula **=B4*takeup**. Copy it across.
3. In order to calculate revenue, we need to know cumulative customers. Calculate this using the formula **=B5** for **B6** and **=C5+B6** for **C6**.
4. Alternatively, we can use a cumulative sum in cell **B6** which looks like this **=SUM(\$B\$5:B5)**
5. Copy it across

See page 283 of Using Excel for Business Analysis for more information on building Cumulative Totals

Logic error: We've missed a critical step! By calculating the customer numbers in this way, we are grossly over-estimating the number of customers – particularly in the first couple of years.

Can you work out where we've gone wrong?

Fix up your model, and write down the answer here:

Video 5.4: Business Case - Forecasting Customer Revenue

If we calculate the revenue based on this number of customers we'll be *grossly* overestimating the customer number, and hence the revenue. We need to adjust for the *timing* of the customer acquisition.

Watch the video for instructions on how to adjust your customer numbers.

Churn (Customer Defection)

It is unrealistic to expect to keep acquiring customers without losing some, so we need to make some assumptions about churn. We have to expect that people will change their minds and they will drop off. We call it persistency in insurance, churn, lost customers, attrition - whatever you call it, it would be over-ambitious to expect that you are going to keep 100% of your customers.

1. Create a named range called "churn" on the "Assumptions" page and enter a churn amount. Let's go with 10% for now.
2. We may decide to assume that we do not lose any customers for the first two years and churn begins in Year 2. However we have decided to calculate this, you'll need to document this assumption on the "Assumptions" page.
3. We also need to consider how the churn rate is being used. Is it based on the average number of customers, or the year end number? Let's say that we will base it on the year end number of customers.
4. So cell **B8** will be blank, and cell **C8** will contain the formula $=-B6*churn$. We've made it a negative amount so that we can see it's being deducted from the total.
5. Add the average number of customers and the churn amount together in row 9 so that we can then calculate the revenue. Relabel it "Net Average Cumulative Customers" to make it clear what this row contains.
6. Don't forget to document your assumptions. Don't leave it until the end! Go back to the Assumptions page and go to Churn and put the note "based on prior year end balance of customer not average". This way, there can be no possible misinterpretation about what we've put into this financial model.

Premium Calculation

Now that we've determined the assumed number of customers, calculation of the revenue is reasonably straightforward.

1. Create a named range called "premium" on the "Assumptions" page and enter \$400. This doesn't really need any source reference documentation, as this is a key input assumption.
2. Go to the "Revenue" page and calculate the Gross Premiums amount in cell B10 using the formula $=B9*premium$ and copy it across.
3. Index this by inflation – using the same named range "inflation" of course. Use the same compounding method as we did with the staff costs. The formula in cell **B10** should be: $=B9*premium*(1+inflation)^(B2-\$B\$2)$. Copy it across.

4. We now calculate the profit we make on these premiums. Note that we are simply a reseller of this product, so our revenue is only 10% of the premium. Create a named range called "prem_profit" on the "Assumptions" page and enter 10%.
5. Go to the "Revenue" page and calculate the Premium Revenue in cell **B11** using the formula **=B10*prem_profit** and copy it across.
6. Now pull this total through to the "Revenue" line on the P&L page. On the "P&L" page, in cell **B13**, the formula should be: **=Revenue!B11**. Copy it across.
7. We now have our revenue.

Video 5.5: Business Case - Modelling Market Penetration & Expenses

You have also been asked to calculate this products projected market penetration both for HoneyCorp customers and the general population. We're going to calculate this on the top of the P&L page.

Modelling Market Penetration

1. Calculate the Australian Pet-owning Population by multiplying the total Australian population by the Pet Ownership percentage on the "Assumptions" page. Your formula in cell **B7** should look like this: **=Assumptions!B24*pet_ownshp**. Copy it across.
2. Pull the "HoneyCorp Pet-owning Customers" through from where we have already calculated it on the "Revenue" page. In cell **B8**, your formula should be **=Revenue!B4**. Copy it across.
3. Now pull the "Net Cumulative Customers" on the "Revenue" sheet into the "Acquired Customers" field on the "P&L" sheet. In cell **B9** your formula should be **=Revenue!B9**. Copy it across.
4. Calculate the HoneyCorp penetration by dividing the acquired customers by the HoneyCorp pet-owning population using the formula: **=B9/B8**. Copy it across.
5. Calculate the Market penetration by dividing the acquired customers by the total Australian pet-owning population using the formula: **=B9/B7**. Copy it across.

Calculating Fixed & Variable Expenses

1. To calculate overheads, create a named range called "overheads" on the "Assumptions" page and enter \$4.50 per acquired customer.
2. Should we use year-end number of customers, or average? It's up to you – as long as you can justify it, and of course, as long as you document it properly. Let's use average.
3. Now calculate the overheads on the "P&L" page in cell **B16** by multiplying the number of acquired customers by the overhead amount using the formula **=B9*OH**.
4. It would be better practice, however, to link to the source on the Assumptions page instead of where it has been linked through on the Revenue page, hence causing a "spaghetti link" or "daisy chain". Instead, in cell **B16** use the formula **=Revenue!B9*OH**. Copy it across.
5. On the "Assumptions" page add the media & marketing costs and then the product development costs of \$500,000 each in the first year. For the media & marketing costs we are told that they will halve until Year 4, and then remain constant, so enter this accordingly.
6. The formulas on the Assumptions page should look like the below:

26										
27	Media & Marketing	500000	=B27/2	=C27/2	=D27/2	=E27	=F27	=G27	=H27	=I27
28	Product Development	500000								
29										

7. Link these numbers to the "P&L" sheet in rows 17 and 18.

You now have your complete P&L for this product. Your P&L should look something like this:

	Year 1	Year 2	Year 3	Year 4	Year 5	Year 6	Year 7	Year 8	Year 9	Year 10
Market Data										
Australian Pet-owning Population	12,549,447	12,664,057	12,777,791	12,890,539	13,001,957	13,111,913	13,220,270	13,326,873	13,443,043	13,554,196
HoneyCorp Pet-owning Customers	1,188,000	1,198,850	1,209,616	1,220,290	1,230,837	1,241,246	1,251,504	1,261,595	1,272,593	1,283,115
Acquired Customers	14,850	41,716	68,824	96,174	123,762	151,586	179,643	207,928	236,451	265,216
HoneyCorp Penetration	1.3%	3.5%	5.7%	7.9%	10.1%	12.2%	14.4%	16.5%	18.6%	20.7%
Market Penetration	0.1%	0.3%	0.5%	0.7%	1.0%	1.2%	1.4%	1.6%	1.8%	2.0%
Revenue	\$594,000	\$1,718,684	\$2,920,629	\$4,203,682	\$5,571,830	\$7,029,208	\$8,580,110	\$10,228,991	\$11,981,163	\$13,841,863
Staff Costs	\$2,233,200	\$2,035,692	\$2,283,481	\$1,629,912	\$1,678,809	\$1,590,060	\$1,637,762	\$1,686,895	\$1,737,502	\$1,789,627
Overheads	\$66,825	\$187,720	\$309,709	\$432,783	\$556,931	\$682,139	\$808,392	\$935,674	\$1,064,030	\$1,193,471
Media & Marketing	\$500,000	\$250,000	\$125,000	\$62,500	\$62,500	\$62,500	\$62,500	\$62,500	\$62,500	\$62,500
Product Development	\$500,000	\$0	\$0	\$0	\$0	\$0	\$0	\$0	\$0	\$0
Total Expenses	\$3,300,025	\$2,473,412	\$2,718,191	\$2,125,195	\$2,298,240	\$2,334,699	\$2,508,654	\$2,685,069	\$2,864,031	\$3,045,598
Profit / Loss	-\$2,706,025	-\$754,729	\$202,438	\$2,078,487	\$3,273,590	\$4,694,509	\$6,071,456	\$7,543,922	\$9,117,131	\$10,796,264

Video 5.6: Business Case - Project Evaluation (NPV & IRR)

The most common project evaluation tools are:

1. NPV (Net Present Value)
2. IRR (Internal Rate of Return)
3. Payback period

See the section on “Calculating the WACC” and “Calculating “NPV & IRR” in Module 2, as well as page 174 of Using Excel for Business Analysis for more detailed instructions on using these Project Evaluation tools.

Weighted Average Cost of Capital (WACC) is often calculated to see a company’s cost of capital in order to determine what discount rate, or hurdle rate to use for an NPV calculation.

Although it is interesting to understand the background of how WACC is calculated, we will not need to calculate it for this model, as we have been given an input, or “house rate” of 12%.

Our NPV determines the amount of cash left over after the required rate of return (12% in this case) has been satisfied. The IRR is the percentage rate of return that is represented by the series of cashflows in the range.

Although the interpretation of NPV and IRR are complex, the functions are fairly straightforward. Let’s calculate them now.

1. Create a named range “cost_cap” for the cost of capital in cell **B20** on the “Assumptions” page
2. Go back to the “P&L” page and in cell **B24**, calculate the NPV by entering the formula **=NPV(cost_cap,B21:F21)**. Your answer should be \$304,771.
3. Calculate the IRR in cell **B25** by entering the formula **=IRR(B21:F21)**. Your answer should be 15%.

Not bad. So we have a positive NPV which means we should go ahead with the project, right? Maybe, maybe not. Well, the next step will be to do some scenario analysis and we’ll be able to see how reliable we think these project evaluation tools are. But first of all, let’s talk a little more about the meaning of these numbers.

Using a Goal Seek in our Business Case

Because the NPV is how much cash we have left over, and the IRR is the rate of return generated by that cash, we can see that there is a relationship between the two. If the NPV is a positive value, then the rate of return (IRR) must be greater than the cost of capital, right? And if the NPV was zero, then the IRR must be *exactly* the same as the discount rate. Are you sure? Let’s test that.

See the section on “What-if Analysis with Goal Seek” in Module 2, as well as page 247 of Using Excel for Business Analysis for more detailed instructions on using a Goal Seek.

What we want to do is to set our NPV to zero (and test to see if the IRR = the cost of capital in that case). We can set our NPV to zero by changing any of the direct input assumptions. Let's choose the premium price of \$400 in this case. We know that to be accepted, the project must have an IRR of 12%, therefore we want to know what our "slack" in the overheads is.

1. On the "Data" tab, in the "Data Tools" group, click on "What-if Analysis" and select "Goal Seek" from the drop down list.
2. In the Set Cell box, enter the reference for the cell containing the formula producing the end result that you want to achieve (**B5** in this case).
3. In the To Value box, enter the result you would like to achieve in the Set Cell (type in 0 (zero)).
4. In the By Changing Cell, enter the reference on the Assumptions page for the cell that Excel is to change (**B5**, the price input variable).



5. Click OK and this will bring up the Goal Seek Status dialogue box containing the result of your seek.
6. The new value you need to achieve the desired result will be displayed in the changing cell in your model. Your new premium price should be around \$388 which means that we don't have a lot of room for discounting!
7. Change the premium price back to \$400 to get your model back to the way it was

Video 5.7: Business Case - Project Evaluation (Payback Period)

Recommended Reading: See the section on “How to Calculate a Payback Period” in Chapter 9, Page 284 of [Using Excel for Business Analysis](#). Below is an extract from this chapter:

There is no suitable formula in Excel for calculating a payback period (yet!) However, there are a couple of different ways of doing it manually. Demonstrated below are two methods, but this is by no means an exhaustive list. The first method is very simple, but not very accurate, and the second is quite long-winded, but more precise.

Simple Payback Calculation

In a business case summary as shown below, we can see that the payback period (the point at which the sum of our cash flows becomes positive) is between four and five years—at 2022. We can easily estimate this by highlighting the cash totals for each year and looking at the sum amount at the bottom right-hand corner of the page. However, we would like to create a cell that automatically calculates the payback year so that as inputs change due to scenario and sensitivity analysis, the payback year will always be shown in the model, based on the cash flow amounts.

	2018	2019	2020	2021	2022	2023	2024	2025
5								
6	Market Data							
7	Australian Pet-owning Population	12,549,447	12,664,057	12,777,791	12,890,539	13,001,957	13,111,913	13,220,270
8	HoneyCorp Pet-owning Customers	1,188,000	1,198,850	1,209,616	1,220,290	1,230,837	1,241,246	1,251,504
9	Acquired Customers	14,850	41,716	68,824	96,174	123,762	151,586	179,643
10	HoneyCorp Penetration	1.3%	3.5%	5.7%	7.9%	10.1%	12.2%	14.4%
11	Market Penetration	0.1%	0.3%	0.5%	0.7%	1.0%	1.2%	1.4%
12								
13	Revenue	\$575,551	\$1,665,302	\$2,829,915	\$4,073,117	\$5,398,771	\$6,810,883	\$8,313,615
14								
15	Staff Costs	\$2,233,200	\$2,035,692	\$2,283,481	\$1,629,912	\$1,678,809	\$1,590,060	\$1,637,762
16	Overheads	\$66,825	\$187,720	\$309,709	\$432,783	\$556,931	\$682,139	\$808,392
17	Media & Marketing	\$500,000	\$250,000	\$125,000	\$62,500	\$62,500	\$62,500	\$62,500
18	Product Development	\$500,000	\$0	\$0	\$0	\$0	\$0	\$0
19	Total Expenses	\$3,300,025	\$2,473,412	\$2,718,191	\$2,125,195	\$2,298,240	\$2,334,699	\$2,508,654
20								
21	Profit / Loss	-\$2,724,474	-\$808,110	\$111,725	\$1,947,922	\$3,100,531	\$4,476,184	\$5,804,960
22								
23								

- In row 22, create a formula that calculates the cumulative amount of cash. See “Calculating Cumulative Totals” later in this chapter for how to do this. Your formula should be **=SUM(\$B\$21:B21)**.
- In cell **B27**, insert a formula that will return the year in which the cumulative cash amount becomes positive. Remember that we want to look along the row, and find the moment at which it becomes zero or greater. Your formula should be something like this: **=LOOKUP(0,B22:K22,B5:K5)**
- Note that this formula is looking for a close match, which means that it will look along row 22 until it finds zero, and it will return the value before it passes zero.

4. This will give you the value 2021, but we actually want 2022, so simply add a **+1** to the end of the formula. Your formula should be: **=LOOKUP(0,B22:K22,B5:K5)+1**

Note that this method is only a sample of several different tools and functions that could be used to calculate the payback period. For example, you could use a nested IF function instead, and an INDEX/MATCH combination.

As the numbers for your business case change, so will your payback year. The result is only a rough estimate. For a more specific (albeit theoretical) payback date, we can use a more complex calculation, as shown by the following.

More Complex Payback Calculation

If you want to see exactly the year and month at which (theoretically) your project turns a profit, there is a more complex method. This way is also quite straightforward, but a little more time-consuming.

1. Go to the “Payback” sheet where the months and years have already been laid out for you.
2. In column A, insert the years that correspond with the months. In this case, the first year is 2018, so enter 2018 into cell **A2**. Copy this down, and change it to 2019 after 12 months. This can be most easily done by entering the formula **=A1** into cell A14, and copying it right down to the bottom.
3. We are going to use column A as the criteria to pick up the cash in each year. In column B, enter a formula that will reference the “P&L” page, and return the cash for that year, which should be divided by 12 to show the cash for each month. Your formula in cell B2 should be **=HLOOKUP(A1,'P&L'!\$B\$5:\$K\$21,17,0)/12**. This is (theoretically) how much cash the project has cost after one month of operation. We are assuming, in this formula, that the sheet that contains your cash flow is called P&L.
4. Then make this a cumulative amount in column B, by changing the formula in Month 2 so that it adds to the previous total. Your formula in cell **B2** should be **=HLOOKUP(A2,'P&L'!\$B\$5:\$K\$21,17,0)/12+B1**. Copy this down the column.

Unfortunately the cumulative SUM calculation that was performed in the last exercise will not work in this instance, so we need to create an inconsistent formula. If this really bothers you, you could add a row and put a zero in cell **B2** instead, so that the formulas are consistent.

6. Continue this calculation right down the page until you reach Year 7 or 8, or as far as you want your payback calculations to work. Remember that if you're doing sensitivity and scenario on this later, you don't want the formula to return an error if the calculations can't handle a larger value, so go down a little bit further than you think you'll need.
7. You can now perform a close match VLOOKUP or LOOKUP, which will return the year and month at which the return becomes positive. This is the payback period.
8. Enter the formula **=VLOOKUP(0,Payback!B1:C120,2,TRUE)** into cell **B28** on the “P&L” sheet. This will return the exact month and year that the project will (theoretically, of course) become profitable.

Tip: There is some debate as to whether payback period should be discounted or not. I generally do not discount payback periods (as shown in this example), but whether or not it has been discounted needs to be stated clearly in your assumptions.

9. This will return 4 Years and 4 Months which is not entirely accurate, as we know that the close match VLOOKUP will always force the result back one cell. So you can change the “Payback” sheet so that the timing starts from 2 months instead of 1 month, which will change the payback period to a more accurate value of 4 Years and 5 Months.

Important Notice: *These course materials are protected by copyright. They are provided to students of Plum Solutions Online Training courses in soft PDF format and so may be printed for study purposes by the licensed student, but may not be copied, redistributed, posted to a website or reproduced in any way.*